* CRITERI d'ARRESTO *

- siano $[a,b]$, h che verificano le ip (1) e (2') del Teo di convergenza, α punto unito di h in $[a,b]$, x_k una succes generata dal m. it def da h con $x_k \in [a,b]$, $x_k \to \alpha$.

$\boxed{1}$ se $|x_{k+1} - x_k| < tol$ allora STOP

- è calcolabile e efficace $(x_{k+1} - x_k \to 0 \ldots)$

- $$x_{k+1} - x_k = h(x_k) - x_k =$$
$$= (h(x_k) - h(\alpha)) - (x_k - \alpha)$$
$$= (h'(\theta) - 1)(x_k - \alpha)$$
con θ tra x_k e α

$$\Rightarrow |x_{k+1} - x_k| = \underbrace{|1 - h'(\theta)|}\,|x_k - \alpha|$$

$$= 1 - h'(\theta) = 1 + \varepsilon_k \quad \text{con} \quad |\varepsilon_k| \leq L < 1$$

Posto: $|x_k - \alpha| = d_k$, $|x_{k+1} - x_k| = \Delta_k$

si riscrive: $\Delta_k = (1+\varepsilon_k)\, d_k$

$$\text{con} \quad \lim_{k\to\infty} |\varepsilon_k| = |-h'(\alpha)| < 1$$

Per k grande, Δ_k approssima d_k con errore relativo $\simeq -h'(\alpha)$.

Es: $|h'(\alpha)| = \frac{1}{100}$, $d_k = 10^{-4}$

$$\Rightarrow \Delta_k \in 10^{-4} \pm 10^{-6} \quad (\text{bene: } \Delta_k \approx d_k)$$

<u>Ma</u> se $|h'(\alpha)| = \frac{9}{10}$, $d_k = 10^{-4}$

$$\Rightarrow \Delta_k \in [0{,}1\ 10^{-4};\ 1{,}9\ 10^{-4}]$$

(meno bene: Δ_k può essere significativamente più piccolo o più grande di d_k!)

$\boxed{2}$ <u>se</u> $\left| \frac{x_{k+1} - x_k}{1 - h'(x_k)} \right| < tol$ <u>allora</u> STOP

- è calcolabile e efficace $\left(\frac{x_{k+1} - x_k}{1 - h'(x_k)} \to 0 \ldots \right)$

- $\frac{x_{k+1} - x_k}{1 - h'(x_k)} = \frac{h'(\theta) - 1}{1 - h'(x_k)} (x_k - \alpha)$ con θ tra...

- posto $\left|\dfrac{x_{k+1} - x_k}{1 - h'(x_k)}\right| = F_k$ si ha :

$$F_k = \underbrace{\left|\frac{1-h'(\theta)}{1-h'(x_k)}\right|} \; d_k$$

$$= \frac{1-h'(\theta)}{1-h'(x_k)} = 1 + \varepsilon_k$$

con $\varepsilon_k = \dfrac{h'(x_k) - h'(\theta)}{1 - h'(x_k)} \xrightarrow{k \to \infty} 0$

Per k grande, F_k approssima d_k con errore relativo $\to 0$!

```
//
// LMV_MetodiUnPunto_00.sce (reperibile sulla pagina web del corso)
//
// Metodo ad un punto definito dalla funzione h (con derivata continua)
//
function y=h(x)
    y = exp(-x)
endfunction
//
function j=dh(x)
    j = -exp(-x)
endfunction
//
// Grafici iniziali
//
clf();
subplot(211);
xx = linspace(-1,3,200);
plot(xx,h(xx),"b",xx,xx,"k");
xlabel("x");
legend("y = h(x)","y = x");
xgrid();
subplot(212);
plot(xx,abs(dh(xx)),"b");
xlabel("x");
ylabel("|h''(x)|")
xgrid();
//
x0 = input("scelta punto iniziale: x0=");
//
// Iterazione
//
x_mit = x0;
tol = 1d-6; // Errore assoluto richiesto.
kmax = 30;
StimaErr = []; // Riga delle stime di errore.
info = -1; // Flag per decidere se interrompere l'iterazione.
k = 0;
//
while info == -1,
    if k > kmax then info = 3; z = x_mit; break; end;
    if dh(x_mit) == 1 then info = 2; z = []; end;
    StimaErr($+1) = abs((h(x_mit) - x_mit)/(dh(x_mit) - 1));
        // StimaErr: vedi nota finale.
    if StimaErr($) < tol then info = 1; z = x_mit; break; end;
    x_mit = h(x_mit);
    k = k+1;
end;
//
// Fine iterazione
//
printf("\nz ="); disp(z);
printf("\nnumero iterazioni ="); disp(k);
printf("\ninfo ="); disp(info);
//
// Grafici
//
clf();
subplot(121);
plot(xx,h(xx),"b",xx,xx,"k",z,h(z),"ro");
xlabel("x");
legend("y = h(x)","y = x","(z,h(z))");
```

```
xgrid();
subplot(122);
plot2d(log10(StimaErr(1:$-1)),log10(StimaErr(2:$)),5,frameflag=4);
xlabel("log(e(k))");
ylabel("log(e(k+1))");
xgrid();
//
// Se h ha derivata prima continua allora, detto P il punto unito di h:
//
//   h(x_mit) - x_mit    h(x_mit) - h(P) - (x_mit - P)
//   ----------------- =  ------------------------------ =
//    h'(x_mit) - 1              h'(x_mit) - 1
//
//       h'(t)(x_mit - P) - (x_mit - P)     (h'(t) - 1) (x_mit - P)
//     = ------------------------------- =  ------------------------
//               h'(x_mit) - 1                   h'(x_mit) - 1
//
// con t tra x_mit e P (Teorema di Lagrange). SE la successione
// generata converge a P, allora anche t -> P e:
//
//             h'(t) - 1
//           -------------  ->  1
//           h'(x_mit) - 1
//
// Quando x_mit è vicino a P, lo stimatore fornisce valori affidabili.
//
```

Esercizio : sia $h(x) = x^2 : \mathbb{R} \to \mathbb{R}$

- determ i pti uniti di h (siano α_0, α_1);
- determ $h'(\alpha_0)$, $h'(\alpha_1)$ ed utilizz per decidere se il m. it def da h sia utilizz per appross α_0 ed α_1;
- modif il file di esempio relativo ai metodi ad un punto (LMV_MetodiUnPunto_00.sce) per utilizzarlo con la funz h assegnata
  (N.B: utilizz opportunam l' operatore .^ :
  se x è una matrice $r \times s$, allora
  x.^2 = la matrice $r \times s$ di elem i,j
  dato da x_{ij}^2);
- Scegliere $x_0 = 0.45$ ed interpretare il grafico finale riguardante la stima dell' errore;
- Scegliere $x_0 = 0,5$ e giustificare l' errore dichiarato dalla procedura;
- Scegliere $x_0 = -1$ ed interpretare i risultati (l' iterazione termina con successo, ma poi nasce un problema nel grafico finale ...)
- Scegliere $x_0 = -0.999$ e poi $x_0 = -1,001$:

accade qualcosa di "imprevisto" ?

- determ tutti i valori x_0 t.c. la succes generata dal m. it def da h a partire da x_0 converge a 1.